KIT
de sobrevivencia em
CALCULO
Prof. Doherty Andrade - DMA- UEM
doherty@gauss.dma.uem.br

Extremos Locais e Globais

Nesta seção o problema é encontrar os extremos de funções de mais de uma variavel. Considere a função definida por:

$$f(x,y) = x\,y\,e^{\left(-\frac{x^2+y^2}{2}\right)}$$.

A função pode ser definida pelo Maple V pelo seguinte comando.

> **f := (x,y) -> x*y*exp(-(x^2+y^2)/2);**

$$f := (x,y) \to x\,y\,e^{\left(-\frac{1}{2}x^2 - \frac{1}{2}y^2\right)}$$

Para ter uma ideia dos valores extremos podemos plotar o grafico. Esta função se comporta como

> **plot3d(f(x,y),x=-3..3,y=-3..3,style =patch,orientation=[70,65],**

> **axes=FRAMED);**



Parece que temos um extremo em cada quadrante. Existe um minimo no segundo e no quarto quadrante e um maximo no primeiro e no terceiro quadrante. Um desenho com curvas de nivel pode ser util para encontrar uma localizacao aproximada dos pontos criticos.

> **with(plots):**

> **contourplot(f(x,y),x=-3..3,y=-3..3);**



Para encontrar a localização exata dos pontos criticos de f precisamos determinar as primeiras derivada.

```
> fx := diff(f(x,y),x);
```

$$fx := y\,\mathbf{e}^{\left(-\frac{1}{2}x^2-\frac{1}{2}y^2\right)} - x^2\,y\,\mathbf{e}^{\left(-\frac{1}{2}x^2-\frac{1}{2}y^2\right)}$$

A expressão f_x pode ser simplificada.

```
> fx := factor(fx);
```

$$fx := -y\,\mathbf{e}^{\left(-\frac{1}{2}x^2-\frac{1}{2}y^2\right)}(x-1)(x+1)$$

Faremos o mesmo para a derivada com relação y.

```
> fy := factor(diff(f(x,y),y));
```

$$fy := -x\,\mathbf{e}^{\left(-\frac{1}{2}x^2-\frac{1}{2}y^2\right)}(y-1)(y+1)$$

Os pontos criticos sao os pontos onde f_x e f_y se anulam simultaneamente.

```
> SOL := solve({fx=0,fy=0},{x,y});
```

$$SOL := \{x=0, y=0\}, \{y=1, x=1\}, \{y=-1, x=1\}, \{x=-1, y=1\}, \{x=-1, y=-1\}$$

Segue que existem cinco pontos criticos:

(0,0), (1,1), (1,-1), (-1,1), e (-1,-1).

Inspecionando o desenho feito acima vemos que os pontos (1,1) e (-1,-1) são pontos de máximo, e os pontos (-1,1), (1,-1) são de minimo. Finalmente, o desenho também sugere que (0,0) é um ponto de sela.

Para ver melhor o desenho, vamos ver cada ponto critico mais de perto. Primeiro checamos na vizinhança de (0,0).

```
> plot3d(f(x,y),x=-0.1..0.1,y=-0.1..0.1,orientation = [70,65],
> style=patchcontour);
```



>

De fato este desenho evidencia que o ponto (0,0) é um ponto de sela. Agora checamos a vizinhança de (1,1) que antes parecia ser de máximo.

> **plot3d(f(x,y),x=0.9..1.1,y=0.9..1.1,orientation = [70,65],**

> **style=patchcontour);**



Desta maneira podemos obter uma evidencia clara plotando uma vizinhança dos outros pontos criticos. Mas o melhor criterio eum um criterio analitico para caracterizar pontos criticos.

Lembramos que se (a,b) é um ponto critico de f

$\Delta = f_{xx}(a,b) f_{yy}(a,b) - f_{xy}(a,b)^2$,

então

1. Se Δ > 0 e $f_{xx}(a,b)$ > 0, (a,b) é um minimo.

2. Se Δ > 0 e $f_{xx}(a,b)$ < 0, (a,b) é um maximo.

3. Se Δ < 0, então (a,b) é um ponto de sela.

4. Se Δ = 0, tudo pode acontecer.

> **fxx := factor(diff(f(x,y),x,x));**

$$fxx := x\, y\, \mathbf{e}^{\left(-\frac{1}{2}x^2 - \frac{1}{2}y^2\right)} (-3 + x^2)$$

> **fxy := factor(diff(f(x,y),x,y));**

$$fxy := e^{\left(-\frac{1}{2}x^2-\frac{1}{2}y^2\right)}(y-1)(y+1)(x-1)(x+1)$$

> **fyy := factor(diff(f(x,y),y,y));**

$$fyy := x\,y\,e^{\left(-\frac{1}{2}x^2-\frac{1}{2}y^2\right)}(-3+y^2)$$

> **Delta := factor(fxx*fyy-fxy^2);**

$$\Delta := -\left(e^{\left(-\frac{1}{2}x^2-\frac{1}{2}y^2\right)}\right)^2(-5\,x^2\,y^2+y^4\,x^2+x^4\,y^2+1-2\,x^2+y^4-2\,y^2+x^4)$$

Agora vamos aplicar o teste usando o ponto critico (0,0):

> **eval(subs(x=0,y=0,Delta));**

$$-1$$

Assim Δ < 0 no ponto (0,0) e este ponto é um ponto de sela.

Agora checar os valores de f_{xx} e Δ no ponto (1,1).

> **subs(x=1,y=1,[fxx,Delta]);**

$$\left[-2\,e^{(-1)},\,4\,\left(e^{(-1)}\right)^2\right]$$

Como o numero **e** é positivo, o resultado significa que f_{xx} < 0, e Δ > 0 em (1,1), o que prova que (1,1) é um ponto onde o maximo ocorre. Podemos usar o *evalf* para aproximar os valores.

> **evalf(");**

$$[-.7357588824,\ .5413411332]$$

A seguir verificamos o ponto critico (-1,1).

> **subs(x=-1,y=1,[fxx,Delta]);**

$$\left[2\,e^{(-1)},\,4\,\left(e^{(-1)}\right)^2\right]$$

Assim (-1,1) é um minimo.

Finalmente para (-1,-1) e (1,-1).

> **subs(x=-1,y=-1,[fxx,Delta]);subs(x=1,y=-1,[fxx,Delta]);**

$$\left[-2\,e^{(-1)},\,4\,(e^{(-1)})^2\right]$$

$$\left[2\,e^{(-1)},\,4\,(e^{(-1)})^2\right]$$

Voce pode concluir que (-1,-1) é um maximo e (1,-1) é um minimo.

O seguinte segmento do Maple V mostra como obter os valores de f(x,y), usamos o comando **zip** :

> **zip(f,[1,-1,-1,1],[1,1,-1,-1]);**

$$[e^{(-1)}, -e^{(-1)}, e^{(-1)}, -e^{(-1)}]$$

Os valores extemos aproximados são:

> **evalf(",3);**

$$[.368, -.368, .368, -.368]$$

>

<u>Resumo</u>: $x\,y\,e^{\left(-\frac{x^2+y^2}{2}\right)}$

(a,b) $f_{xx}(a,b)$ Δ Conclusão $f(x,y)$

(0,0) 0 Negativo Ponto de sela 0

(1,1) Negativo Positivo Maximo .368

(-1,1) Positivo Positivo Minimo -.368

(-1,-1) Negativo Positivo Maximo .368

(1,-1) Positivo Positivo Minimo -.368

Exercicios

Determine e classifique todos os pontos criticos das seguintes funções de duas variáveis. Se um extremo é absoluto então explique porque.

1. $f(x,y) = x\,y - x^2 - y^2 - 2\,x - 2\,y + 4$

2. $f(x,y) = \frac{x^3}{3} + 9\,y^3 - 4\,x\,y$

3. $f(x,y) = \mathbf{e}^{(-2\,x)} \cos(y)$

4. $f(x,y) = x\,y + 2\,x - \ln(x^2\,y)$, $0 < x$, $0 < y$